

## **USING MC68HC05F6 AS TONE PULSE DIALER WITH MELODY-ON-HOLD**

Prepared by Derek Tam and Michael Chang  
Motorola Semiconductors Hong Kong Ltd.

### **INTRODUCTION**

The MC68HC05F6 is a fully static single chip CMOS Microcomputer. It has 288 bytes of RAM and 32 bytes of stack, 4608 bytes of ROM, on-chip oscillator, 4 I/O ports and also features a DTMF / Melody Generator (DMG), a high current port, keyboard interrupt port, Serial Peripheral Interface (SPI), 16-bit timer and a CPU core with two and four software selectable operating speeds respectively. Put simply, the MC68HC05F6 is specifically designed for telephone applications such as high end feature phones or telephone answering machines.

This application note illustrates a simple hardware and software implementation for using the MC68HC05F6 to achieve DTMF dialing, pulse dialing, pacifier tone, and melody-on-hold functions.

### **DTMF / MELODY GENERATOR**

The DMG is a multi-functional tone generator which supports DTMF, melody, and pacifier tone generation with user-selectable frequencies. The benefit of the DMG lies in its versatility. It has sine/square wave melody and pacifier tone generators built-in, thus eliminating a significant amount of external hardware.

As shown in Figure 1, DMG is basically a DTMF generator with row and column frequency dividers, memory scanners, memories for sine wave, resistor ladders for D/A conversion, and a current summer/low-pass active filter. The time base for this module is a 3.58MHz/2 clock source which is derived from the crystal oscillator. The row and column Frequency dividers divide this clock into the lower frequencies according to the number written into Row and Column Frequency Control Registers. The output of each frequency divider drives a memory scanner that sequentially scans one of the 28 locations of a sine wave memory. The frequencies for dual tone melody are also generated by this principle. A D/A converter, which has a built-in active low-pass filter, sums the currents generated by the row and the column resistor ladders. There is also a software option for square wave melody output which is achieved by masking the least five significant bits of the output of the sine wave memory. The pacifier tone is the

signal tapped from the most significant bit of the output of Row Sine Wave Memory.

### **KEYBOARD INTERRUPT**

Port A can be configured as a keyboard interrupt port by using software option. In the keyboard interrupt mode, this port is internally pulled high, thus a keyboard interrupt will be recognised when one of its inputs is externally pulled low.

### **SPEECH CIRCUIT**

MC34114 is a monolithic integrated telephone speech network which incorporates the functions of transmit/receive amplification and sidetone control, each with externally adjustable gain. Loop length equalization varies the gains based on loop current. The microphone amplifier has a differential input stage designed to reduce RFI and common mode noise pick up. A MUTE\* input mutes the microphone and receive amplifiers during dialing. A regulated output voltage is provided for biasing of the microphone, and a separate output voltage powers an external microprocessor. The MC34114 is designed to operate at a minimum of 1.2 volts, making party line operation possible.

### **DESCRIPTION OF THE TELEPHONE SYSTEM**

MC68HC05F6 is designed not only for performing telephone dialing functions but also for interfacing with external peripheral, for example, a telephone answering machine. However, only the former application, but not the latter is described in this application note. Thus, many I/O lines, which should be dedicated for controlling functions, are intentionally used to drive a simple 7-segment LED display.

### **HARDWARE AND FUNCTIONAL DESCRIPTION**

As shown in Figure 2, the telephone system consists of the MC68HC05F6, a 4x4 keyboard, two 7-segment LED displays, speech circuit MC34114P, telephone ringer MC34017-1, rectifier bridge and a transistorized pulse di-



The two 7-segment LEDs are used to display the key number pressed. Note that key 'A' is displayed as '10', key '#' as '14' and key '\*' as '15'.

Either pulse or DTMF dialing can be selected through switch S3. Square wave or sine wave melody can be output by pressing '\*' and '#' after switching on melody switch S4 during a conversation. A number is dialed by pressing the appropriate key, except that pressing numbers 'A' to 'D' and '\*' and '#' renders no response for pulse dialing.

emitter and the other, a common base configuration. The TCMP, which is the output compare pin of the 16-bit timer, is set high before and after pulse dialing to bias the base of TR1. Thus, the current flows through R3 and turns on TR2, and consequently causes the loop current of about 10mA to flow. When the TCMP pin is low, transistors TR1 and TR2 will be turned off, causing the loop current to decrease. The pulse train at TCMP will therefore effect pulse dialing across tip and ring. After dialing, the speech circuit will be un-muted.

## SOFTWARE

Since the program is well-documented, please refer to the program listing for details.





```

00004 *****
00005 *
00006 *   APPLICATION PROGRAM
00007 *   FOR THE MC68HC05F6
00008 *
00009 *   by Michael Chang
00010 *
00011 *****
00012 *
00013 *   I/O REGISTER DEFINITIONS
00014 *
00015 0000 0000 PORTA EQU $00
00016 0000 0001 PORTB EQU $01
00017 0000 0002 PORTC EQU $02
00018 0000 0003 PORTD EQU $03
00019 0000 0004 DDRA EQU $04 DATA DIRECTION REGISTER A
00020 0000 0005 DDRB EQU $05 DATA DIRECTION REGISTER B
00021 0000 0006 DDRC EQU $06 DATA DIRECTION REGISTER C
00022 0000 0007 DDRD EQU $07 DATA DIRECTION REGISTER D
00023 *
00024 *   SPI REGISTERS
00025 *
00026 0000 000a SPCR EQU $0A SERIAL PERIPHERAL CONTROL
00027 0000 000b SPSR EQU $0B SERIAL PERIPHERAL STATUS
00028 0000 000c SPDAT EQU $0C SERIAL PERIPHERAL DATA
00029 0000 0007 SPIF EQU 7 SPI FLAG
00030 0000 0006 DCOL EQU 6 DATA COLLISION FLAG, BIT 6 OF STATUS REGISTER
00031 *
00032 *   DMG REGISTERS
00033 *
00034 0000 000d FCR EQU $0D ROW FREQUENCY CONTROL REGISTER
00035 0000 000e FCC EQU $0E COLUMN FREQUENCY CONTROL REGISTER
00036 0000 000f TNCR EQU $0F TONE CONTROL REGISTER
00037 0000 0007 MS1 EQU 7
00038 0000 0006 MS0 EQU 6
00039 0000 0005 TGER EQU 5 ROW TONE GENERATION ENABLE
00040 0000 0004 TGEC EQU 4 COLUMN TONE GENERATION ENABLE
00041 0000 0000 FR1 EQU $00
00042 0000 0001 FR2 EQU $01
00043 0000 0002 FR3 EQU $02
00044 0000 0003 FR4 EQU $03
00045 0000 0010 FC1 EQU $10
00046 0000 0011 FC2 EQU $11
00047 0000 0012 FC3 EQU $12
00048 0000 0013 FC4 EQU $13
00049 *
00050 *   NOTES DEFINITIONS
00051 *
00052 0000 0004 DS5 EQU $04
00055 0000 0007 FS5 EQU $07
00056 0000 0008 G5 EQU $08
00057 0000 0009 GS5 EQU $09
00058 0000 000a A5 EQU $0A
00059 0000 000b AS5 EQU $0B
00060 0000 000c B5 EQU $0C
00061 0000 000d C6 EQU $0D
00062 0000 000e CS6 EQU $0E
00063 0000 000f D6 EQU $0F
00064 0000 0014 DS6 EQU $14
00065 0000 0015 E6 EQU $15
00066 0000 0016 F6 EQU $16

```

00067	0000	0017	FS6	EQU	\$17	
00068	0000	0018	G6	EQU	\$18	
00069	0000	0019	GS6	EQU	\$19	
00070	0000	001a	A6	EQU	\$1A	
00071	0000	001b	AS6	EQU	\$1B	
00072	0000	001c	B6	EQU	\$1C	
00073	0000	001d	C7	EQU	\$1D	
00074	0000	001e	CS7	EQU	\$1E	
00075	0000	001f	D7	EQU	\$1F	
00076	0000	0009	DLY1	EQU	\$09	
00077	0000	0013	DLY2	EQU	\$13	
00078	0000	003b	DLY6	EQU	\$3B	
00079		*				
00080		*	MISCELLANEOUS DEFINITIONS			
00081		*				
00082	0000	0010	EVENR	EQU	\$10	EVENT ENABLE REGISTER
00083	0000	0011	MISCR	EQU	\$11	MISCELLANEOUS REGISTER
00084	0000	0004	KEYF	EQU	4	
00085		*				
00086		*	TIMER REGISTERS			
00087		*				
00088	0000	0012	TIMCR	EQU	\$12	TIMER CONTROL
00089	0000	0013	TIMSR	EQU	\$13	TIMER STATUS
00090	0000	0014	IPCAPH	EQU	\$14	INPUT CAPTURE (HIGH BYTE)
00091	0000	0015	IPCAPL	EQU	\$15	INPUT CAPTURE (LOW BYTE)
00092	0000	0016	OPCOMH	EQU	\$16	OUTPUT CAPTURE (HIGH BYTE)
00093	0000	0017	OPCOML	EQU	\$17	OUTPUT CAPTURE (LOW BYTE)
00094	0000	0018	COUNTH	EQU	\$18	COUNTER (HIGH BYTE)
00095	0000	0019	COUNTL	EQU	\$19	COUNTER (LOW BYTE)
00096	0000	001a	DUALTH	EQU	\$1A	DUAL TM COUNTER (HIGH BYTE)
00097	0000	001b	DUALTL	EQU	\$1B	DUAL TM COUNTER (LOW BYTE)
00098	0000	0006	OCF	EQU	6	OUTPUT COMPARE FLAG, BIT 6
00099	0000	0006	OCE	EQU	6	OUTPUT COMPARE ENANBLE, BIT 6
00100	0000	0000	OLVL	EQU	0	OUTPUT LEVEL AT TCMP OUTPUT
00101	0000	001f	KEYCR	EQU	\$1F	KEYBOARD INTERRUPT REGISTER
00102	0000	002c	SYSOP	EQU	\$2C	CLOCK OPTION REGISTER
00103	0000	0006	INTE1	EQU	6	IRQ1 ENABLE, BIT 6
00104	0000	0004	INTN1	EQU	4	IRQ1 EDGE SENSITIVITY, BIT 4
00105	0000	0006	INTF1	EQU	6	IRQ1 INTERRUPT FLAG, BIT 6
00106	0000	0007	KEYE	EQU	7	KEYBOARD INTERRUPT ENABLE, BIT 7
00107	0000	0001	KEYX1	EQU	1	KEYBOARD INTERRUPT EXTEND 1, BIT 1
00108	0000	0000	KEYX0	EQU	0	KEYBOARD INTERRUPT EXTEND 0, BIT 0
00109	0000	0007	SCS1	EQU	7	SYSTEM CLOCK SELECT 1, BIT 7
00110	0000	0006	SCS0	EQU	6	SYSTEM CLOCK SELECT 0, BIT 6
00111	0000	0005	TCS	EQU	5	TIMER CLOCK SELECT, BIT 5
00112		*				
00113		*	OPTION DEFINITIONS			
00114		*	(DEFINED BY PB4-5 INPUT PINS STATUS)			
00115		*				
00116	0000	0005	DILOP	EQU	5	DIALING OPTION
00117	0000	0004	FUNOP	EQU	4	FUNCTION OPTION
00118		*				
00119		*	SPEECH CCT CONTROL PIN DEFINITIONS			
00120		*	(DEFINED BY PA4-5 OUTPUT PIN STATUS)			
00121		*				
00122	0000	0004	MUTEB	EQU	4	MUTE CONTROL (1: NOT MUTE)
00123	0000	0005	DILMD	EQU	5	DIALING MODE (0:DTMF 1:PULSE)
00124		*				
00125		*	MEMORY MAP			
00126		*				
00127	0000	0030	COUNT	EQU	\$30	TIMER HIGH BYTE FOR DELAY SUBROUTINE

```

00128 0000 0031 KEYPOS EQU $31 KEY POSTION CODED BY COUNTING ROW AND COLUMN
00129 0000 0032 KEYNO EQU $32 KEY NUMBER ON KEY PAD
00130 0000 0033 KEYROW EQU $33 KEY ROW NUMBER
00131 0000 0034 KEYCOL EQU $34 KEY COLUMN NUMBER
00132 0000 0035 ROUND EQU $35 ROUND NUMBER OF MUSICAL PIECE
00133 0000 0036 HTEMP EQU $36 TEMPORARY LOCATION TO STORE HIGH BYTE
00134 *
00135 0000 0042 MUL EQU $42 MULTIPLY INSTRUCTION
00136 *
00137 *****
00138 *
00139 * INITIALISATION *
00140 *
00141 * BUS clock : E *
00142 * Timer clock : E/8 *
00143 * PORT A : PA0-3 (INPUT) FOR KEYBOARD SENSING *
00144 * PA4=1 (NOT MUTE) PA5=0 (PULSE MODE) *
00145 * PORT B : PB0-3 (OUTPUT ALL '0') FOR KEYBOARD SCAN *
00146 * PB4-5 (INPUT FOR OPTIONS) *
00147 * PORT C : OUTPUT (OUTPUT ALL '1' ie. segment off) *
00148 * PORT D : OUTPUT (OUTPUT ALL '1' ie. segment off) *
00149 *
00150 *****
00151 *
00152 0d00 ORG $D00
00153 0d00 a6 e0 START LDA #%11100000 SET MBUS CLOCK = E TIMER CLOCK = E/8
00154 0d02 b7 2c STA SYSOP
00155 0d04 a6 7f INT1 LDA #$7F
00156 0d06 b7 02 STA PORTC DISPLAY DOT IN MSB
00157 0d08 b7 03 STA PORTD DISPLAY DOT IN LSB
00158 0d0a 33 06 COM DDRC INIT PORTC and PORTD
00159 0d0c 33 07 COM DDRD TO DRIVE COMMON ANODE 7-SEG LED
00160 0d0e a6 10 LDA #%00010000
00161 0d10 b7 00 STA PORTA SET NORMAL SPEECH MODE
00162 0d12 a6 f0 LDA #$F0
00163 0d14 b7 04 STA DDRA
00164 0d16 3f 01 CLR PORTB
00165 0d18 43 COMA
00166 0d19 b7 05 STA DDRB
00167 *****
00168 *
00169 * -- STARTUP DISLAY -- *
00170 * THE DISPLAY WILL PERFORM A COUNT DOWN SEQUENCE FROM 9 TO 0 ON *
00171 * BOTH LED WHICH THE TIME INTERVAL IS ONE SECOND. AFTERWARDS *
00172 * THE KEYBOARD EXPECTS KEY PRESS FOR DIALING MODE OR MELODY. *
00173 *
00174 *****
00175 *
00176 0d1b ae 09 INTRO LDX #9
00177 0d1d a6 14 INTRO1 LDA #20 DELAY FOR 1 S
00178 0d1f b7 30 STA COUNT
00179 0d21 cd 0dc1 DEL2 JSR DELAY
00180 0d24 3a 30 DEC COUNT DELAY FOR .05S
00181 0d26 26 f9 0d21 BNE DEL2
00182 0d28 d6 0e55 LDA TBSEG,X FETCH SEGMENT TABLE FOR DIGIT
00183 0d2b b7 03 STA PORTD DISPLAY LSB
00184 0d2d b7 02 STA PORTC DISPLAY MSB
00185 0d2f 5a DECX .. TIMER CLOCK
00186 0d30 2a eb 0d1d BPL INTRO1
00187 *
00188 0d32 ae 09 LDX #$09 TO INITIALIZE TCMP PIN FOR DTMF AND
00189 0d34 b6 12 INTCMP LDA TIMCR PULSE DIALING. H/W NEED.

```

```

00190 0d36 a8 01      EOR  #%00000001
00191 0d38 b7 12      STA  TIMCR
00192 0d3a cd 0dc1     JSR  DELAY
00193 0d3d 5a          DECX
00194 0d3e 26 f4 0d34  BNE  INTCMP

```

```

00195 *****

```

```

00196 *
00197 *   KEYBOARD SCANNING AND MAIN PROGRAM
00198 *   When a keyboard interrupt is present, the pressed key will
00199 *   be detected by scanning the keyboard. The key scan results
00200 *   for row and column are stored and then decoded into key
00201 *   position and number. After that, the system will perform DTMF
00202 *   dialing, pulse dialing or melody according to the status on
00203 *   PB4-5. For DTMF dialing and pacifier tone for pulse dialing,
00204 *   the tone will be output as long as the key held. For melody
00205 *   performance, sine wave melody id generated if '*' is pressed,
00206 *   and square wave if '#' is pressed; other keys will be ignored.
00207 *

```

C	8	4	0
D	9	5	1
E	A	6	2
F	B	7	3

KEYPOS

1	2	3	A
4	5	6	B
7	8	9	C
F	0	E	D

KEYNO

```

00208 *****

```

```

00209 0d40 1e 1f      KEYSUB      BSET KEYE,KEYCR      ENABLE KEYBOARD IRQ
00210 0d42 9a          CLI
00211 0d43 8e          STOP
00212 0d44 20 fd 0d43  BRA  STOP
00213 0d46 19 11      KEYSCHN      BCLR KEYF,MISCR
00214 0d48 a6 f7      LDA  #SF7      SCAN FROM 4TH COLUMN
00215 0d4a b7 01      STA  PORTB
00216 0d4c b6 00      REPEAT      LDA  PORTA      DETECT IF KEYPRESSED
00217 0d4e a4 0f      AND  #$0F
00218 0d50 a1 0f      CMP  #$0F
00219 0d52 26 07 0d5b BNE  GOTIT
00220 0d54 34 01      LSR  PORTB      TRY NEXT COLUMN
00221 0d56 25 f4 0d4c BCS  REPEAT
00222 0d58 3f 01      RETURN      CLR  PORTB      PREPARE FOR NEXT SCAN
00223 0d5a 80          RTI
00224 *
00225 0d5b b6 01      GOTIT      LDA  PORTB
00226 0d5d be 00      LDX  PORTA
00227 0d5f b7 34      STA  KEYCOL      SAVE COLUMN POSITION
00228 0d61 bf 33      STX  KEYROW      SAVE ROW POSITION
00229 0d63 ad 56 0dbb BSR  DBOUNC      CHECK FOR NOISE
00230 0d65 b3 00      CPX  PORTA
00231 0d67 26 ef 0d58 BNE  RETURN
00232 0d69 cd 0ddf      JSR  DECCOL      DECODE KEY
00233 0d6c 08 0113 0d82 BRSET FUNOP,PORTB,DIAL1      **** ACTION FOR MELODY
00234 0d6f ad 39 0daa BSR  RELEAS      WAIT FOR KEY RELEASE
00235 0d71 b6 32      LDA  KEYNO
00236 0d73 a4 0e      AND  #$0E      CHECK IF VALID MELODY ACTIVATE KEY
00237 0d75 a1 0e      CMP  #$0E
00238 0d77 26 df 0d58 BNE  RETURN      RETURN IF INVALID
00239 0d79 1b 00      BCLR DILMD,PORTA      SELECT DTMF MODE FOR SPEECH CIRCUIT
00240 0d7b cd 0e79      JSR  MELODY      OUTPUT MUSICAL PIECE
00241 0d7e 3f 0f      CLR  TNCR      TURN OFF TONE GENERATION

```


00242	0d80 20 d6 0d58	BRA RETURN	
00243	0d82 0a 0118 0d9d DIAL1	BRSET DILOP,PORTB,DIAL2	**** ACTION FOR PULSE DIALING
00244	0d85 be 32	LDX KEYNO	
00245	0d87 a3 09	CPX #9	CHECK IF VALID FOR PULSE DIALING
00246	0d89 22 0e 0d99	BHI NOPULS	
00247	0d8b cd 0e29	JSR ECHOTX	ECHO WITH PACIFIER TONE
00248	0d8e ad 1a 0daa	BSR RELEAS	WAIT FOR KEY RELEASE
00249	0d90 3f 0f	CLR TNCR	TURN OFF TONE GENERATION
00250	0d92 1a 00	BSET DILMD,PORTA	SELECT DTMF MODE FOR SPEECH CIRCUIT
00251	0d94 cd 0e33	JSR PUDIAL	PERFORM PULSE DIALING
00252	0d97 20 bf 0d58	BRA RETURN	
00253	0d99 ad 0f 0daa NOPULS	BSR RELEAS	NO PULSE DIALING
00254	0d9b 20 bb0d58	BRA RETURN	
00255	*		
00256	* ACTION FOR TONE DIALING		
00257	*		
00258	0d9d 1b 00 DIAL2	BCLR DILMD,PORTA	SELECT DTMF MODE FOR SPEECH CIRCUIT
00259	0d9f cd 0e22	JSR TNDIAL	PERFORM TONE DIALING
00260	0da2 ad 06 0daa	BSR RELEAS	WAIT FOR KEY RELEASE
00261	0da4 3f 0f	CLR TNCR	TURN OFF TONE GENERATION
00262	0da6 18 00	BSET MUTEb,PORTA	UNMUTE AFTER EACH DIAL
00263	0da8 20 ae 0d58	BRA RETURN	
00264	*		
00265	0daa b6 00 RELEAS	LDA PORTA	CHECK IF KEY RELEASE
00266	0dac aa 0f	ORA #\$0F	
00267	0dae b1 00	CMP PORTA	
00268	0db0 26 f8 0daa	BNE RELEAS	
00269	0db2 ad 07 0dbb	BSR DBOUNC	
00270	0db4 b1 00	CMP PORTA	
00271	0db6 27 f2 0daa	BEQ RELEAS	
00272	0db8 3f 01	CLR PORTB	PREPARE FOR NEXT SCAN
00273	0dba 81	RTS	
00274	*		
00275	0dbb a6 ff DBOUNC	LDA #\$FF	
00276	0dbd 4a AGAIN	DECA	LOOP 1536 TIMES FOR CMOS
00277	0dbe 26 fd 0dbd	BNE AGAIN	
00278	0dc0 81	RTS	
00279	*		
00280	0dc1 1c 12 DELAY	BSET OCE,TIMCR	PROVIDE 0.05S DELAY
00281	0dc3 b6 1a	LDA DUALTH	
00282	0dc5 b7 36	STA HTEMP	
00283	0dc7 a6 ed	LDA #237	
00284	0dc9 bb 1b	ADD DUALTL	
00285	0dcb b7 17	STA OPCOML	
00286	0dcd a6 0a	LDA #10	
00287	0dcf b9 36	ADC HTEMP	
00288	0dd1 b7 16	STA OPCOMH	
00289	0dd3 b6 13	LDA TIMSR	
00290	0dd5 b6 17	LDA OPCOML	DUMMY INSTRUCTION
00291	0dd7 b7 17	STA OPCOML	
00292	0dd9 0d 13fd0dd9 HOLD	BRCLR OCF,TIMSR,HOLD	
00293	0ddc 1d 12	BCLR OCE,TIMCR	
00294	0dde 81	RTS	
00295	*****		
00296	*		*
00297	* DECODE SUBROUTINE		*
00298	* THE KEYS ARE DECODED INTO KEY POSITION(TOP AND RIGHT MOST IS 0		*
00299	* AND BOTTOM AND LEFT MOST IS 15) AND KEY NUMBER (THE NUMBER ON		*
00300	* THE KEYPAD WITH A=11, ... *=14, #=15 etc).		*
00301	*		*
00302	*****		



00303	0ddf 5f	DECCOL	CLR X	
00304	0de0 34 34	DDCOL	LSR KEYCOL	FIND COLUMN DETECTED
00305	0de2 24 03 0de7		BCC COLMUL	
00306	0de4 5c		INC X	COUNT COLUMN
00307	0de5 20 f9 0de0		BRA DDCOL	
00308		*		
00309	0de7 d6 0e4d	COLMUL	LDA TBTONC,X	STORE COLUMN TONE REQ'D
00310	0dea b7 0e		STA FCC	
00311	0dec a6 04		LDA #4	
00312	0dec 42		FCB MUL	MULTIPLY X BY 4
00313	0def b7 31		STA KEYPOS	RESULTS IN A AND STORE TO BUFFER
00314	0df1 34 33	DECROW	LSR KEYROW	FIND ROW DETECTED
00315	0df3 24 05 0dfa		BCC GETSEG	
00316	0df5 5c		INC X	
00317	0df6 3c 31		INC KEYPOS	COUNT ROW AND STORE KEY POSITION
00318	0df8 20 f7 0df1		BRA DECROW	
00319		*		
00320	0dfa d6 0e51	GETSEG	LDA TBTNR,X	STORE ROW TONE REQ'D
00321	0dfd b7 0d		STA FCR	
00322	0dff be 31		LDX KEYPOS	GET KEY POSITION
00323	0e01 d6 0e5f		LDA TBKEY,X	TRANSLATE TO KEY NUMBER
00324	0e04 b7 32		STA KEYNO	
00325	0e06 a0 0a		SUB #10	IS KEY NUMBER > 10 ?
00326	0e08 24 0a 0e14		BHS CARSEG	
00327	0e0a be 32		LDX KEYNO	
00328	0e0c d6 0e55		LDA TBSEG,X	
00329	0e0f b7 03		STA PORTD	
00330	0e11 5f		CLR X	
00331	0e12 20 08 0e1c		BRA OPSEG	
00332	0e14 97	CARSEG	TAX	KEY NUMBER >= 10, CARRY TO MSD
00333	0e15 d6 0e55		LDA TBSEG,X	
00334	0e18 b7 03		STA PORTD	
00335	0e1a ae 01		LDX #1	
00336	0e1c d6 0e55	OPSEG	LDA TBSEG,X	KEY NUMBER <= 10, DISPLAY AS NORMAL
00337	0e1f b7 02		STA PORTC	
00338	0e21 81		RTS	
00339		*		
00340		* TONE DIALING		
00341		*		
00342	0e22 19 00	TNDIAL	BCLR MUTE,PORTA	MUTE, PORTA
00343	0e24 a6 30	TNDIA1	LDA #%00110000	ENABLE DTMF TONE GENERATION
00344	0e26 b7 0f		STA TNCR	
00345	0e28 81		RTS	
00346		*		
00347		* PACIFIER TONE ECHO		
00348		*		
00349	0e29 d6 0e6f	ECHOTX	LDA TBTNX,X	GET PACIFIER TONE
00350	0e2c b7 0d		STA FCR	
00351	0e2e a6 a0		LDA #%10100000	ENABLE PACIFIER TONE GENERATION
00352	0e30 b7 0f		STA TNCR	
00353	0e32 81		RTS	
00354		*		
00355		* PULSE DIALING		
00356		*		
00357	0e33 be 32	PUDIAL	LDX KEYNO	
00358	0e35 26 02 0e39		BNE PDIAL1	
00359	0e37 ae 0a		LDX #10	
00360	0e39 19 00	PDIAL1	BCLR MUTE,PORTA	MUTE SPEECH CIRCUIT
00361	0e3b b6 12	PDIAL2	LDA TIMCR	
00362	0e3d a8 01		EOR #%00000001	TOGGLE OP-COMPARE
00363	0e3f b7 12		STA TIMCR	

00364	0e41 cd	0dc1		JSR DELAY	
00365	0e44 01	12f4 0e3b		BRCLROLVL,TIMCR,PDIAL2	
00366	0e47 5a			DECX	
00367	0e48 26	f1 0e3b		BNE PDIAL2	
00368	0e4a 18	00		BSET MUTEb,PORTA	UNMUTE BEFORE EXIT
00369	0e4c 81			RTS	
00370		*			
00371		*	TABLES FOR THE 7 SEGMENT DRIVE, KEYBOARD NUMBER, AND PACIFIER TONE		
00372		*			
00373	0e4d	13121110	TBTONC	FCB FC4,FC3,FC2,FC1	DTMF COL TONES
00374	0e51	00010203	TBTONR	FCB FR1,FR2,FR3,FR4	DTMF ROW TONES
00375	0e55	c0f9a4b0	TBSEG	FCB \$C0,\$F9,\$A4,\$B0,\$99	7-SEGMENT LEDS
00376	0e5a	9282f880		FCB \$92,\$82,\$F8,\$80,\$90	
00377	0e5f	0a0b0c0d	TBKEY	FCB \$0A,\$0B,\$0C,\$0D	KEYBOARD NUMBERS
00378	0e63	0306090e		FCB \$03,\$06,\$09,\$0E	
00379	0e67	02050800		FCB \$02,\$05,\$08,\$00	
00380	0e6b	0104070f		FCB \$01,\$04,\$07,\$0F	
00381	0e6f	04060809	TBTNX	FCB \$04,\$06,\$08,\$09,\$0B	PACIFIER TONES
00382	0e74	0d0f1416		FCB \$0D,\$0F,\$14,\$16,\$18	
00383		*			
00384		*	MELODY PERFORMANCE : MINUET BY J.S. BACH		
00385		*			
00386	0e79 3f	0f	MELODY	CLR TNCr	
00387	0e7b 01	3202 0e80		BRCLR0,KEYNO,MELOD1	
00388	0e7e a6	40		LDA #01000000	
00389	0e80 b7	0f	MELOD1	STA TNCr	SET DMG MODE
00390	0e82 5f			CLR X	RESET NOTE COUNTER
00391	0e83 3f	35		CLR ROUND	RESET TO FIRST ROUND
00392	0e85 d6	0ed0	ROWTON	LDA MINUET,X	GET ROW NOTE
00393	0e88 27	08 0e92		BEQ COLTON	IF BLANK GO TO COL NOTE
00394	0e8a b1	0d		CMP FCR	CHECK IF SAME NOTE
00395	0e8c 26	02 0e90		BNE ROWTN1	
00396	0e8e 1b	0f		BCLR TGER,TNCr	STOP TONE IF SAME NOTE
00397	0e90 b7	0d	ROWTN1	STA FCR	STORE TONE TO FCR
00398	0e92 5c		COLTON	INCX	INCREMENT NOTE COUNT
00399	0e93 d6	0ed0		LDA MINUET,X	GET COL NOTE
00400	0e96 27	080ea0		BEQ BEAT	IF BLANK GO TO BEAT DELAY
00401	0e98 b1	0e		CMP FCC	CHECK IF SAME NOTE
00402	0e9a 26	020e9e		BNE COLTN1	
00403	0e9c 19	0f		BCLR TGEC,TNCr	STOP TONE IF SAME NOTE
00404	0e9e b7	0e	COLTN1	STA FCC	STORE TONE TO FCC
00405	0ea0 cd	0dc1	BEAT	JSR DELAY	
00406	0ea3 b6	0f		LDA TNCr	TURN ON BOTH PATH
00407	0ea5 aa	30		ORA #00110000	
00408	0ea7 b7	0f		STA TNCr	
00409	0ea9 5c			INCX	INCREMENT NOTE COUNT
00410	0eaa d6	0ed0		LDA MINUET,X	GET BEAT DELAY
00411	0ead b7	30		STA COUNT	
00412	0eaf cd	0dc1	DEL3	JSR DELAY	
00413	0eb2 3a	30		DEC COUNT	
00414	0eb4 26	f9 0caf		BNE DEL3	
00415	0eb6 3d	35		TST ROUND	
00416	0eb8 26	0c0ec6		BNE NXTON1	**** ROUND 0 ****
00417	0eba a3	59		CPX #89	END OF FINE1 ?
00418	0ebc 26	050ec3		BNE NXTON3	NEXT NOTE
00419	0ebe 5f			CLR X	RESET NOTE COUNT
00420	0ebf 33	35		COM ROUND	INCREMENT ROUND COUNT
00421	0ec1 20	c20e85		BRA ROWTON	
00422	0ec3 5c		NXTON3	INCX	
00423	0ec4 20	bf0e85		BRA ROWTON	**** ROUND 1 ****
00424	0ec6 a3	4d	NXTON1	CPX #77	TO FINE2?

00425	0ec8 26	020ecc		BNE	NXTON2	
00426	0eca ae	5f		LDX	#95	GO FINE2
00427	0ecc a3	71	NXTON2	CPX	#113	END OF FINE2 ?
00428	0ece 26	f30ec3		BNE	NXTON3	
00429		*				
00430		*	TABLE FOR DTMF & MELODY			
00431		*				
00432	0ed0	1b0409	MINUET	FCB	AS6,DS5,DLY1 s	
00433	0ed3	040009		FCB	DS5,0,DLY1 d	
00434	0ed6	060009		FCB	F5,0,DLY1 r	
00435	0ed9	080609		FCB	G5,F5,DLY1 m	
00436	0edc	090009		FCB	GS5,0,DLY1 f	
00437	0edf	0b0813		FCB	AS5,G5,DLY2 s	
00438	0ee2	040013		FCB	DS5,0,DLY2 d	
00439	0ee5	040013		FCB	DS5,0,DLY2 d	
00440	0ee8	0d0913		FCB	C6,GS5,DLY2 l	
00441	0eeb	090009		FCB	GS5,0,DLY1 f	
00442	0eee	0b0009		FCB	AS5,0,DLY1 s	
00443	0ef1	180009		FCB	G6,0,DLY1 m	
00444	0ef4	0f0009		FCB	D6,0,DLY1 t	
00445	0ef7	140813		FCB	DS6,G5,DLY2 d2	
00446	0efa	040009		FCB	DS5,0,DLY1 d	
00447	0efd	040009		FCB	DS5,0,DLY1 d	
00448	0f00	090613		FCB	GS5,F5,DLY2 f	
00449	0f03	0b0009		FCB	AS5,0,DLY1 s	
00450	0f06	090009		FCB	GS5,0,DLY1 f	
00451	0f09	080009		FCB	G5,0,DLY1 m	
00452	0f0c	060009		FCB	F5,0,DLY1 r	
00453	0f0f	080413		FCB	G5,DS5,DLY2 m	
00454	0f12	090009		FCB	GS5,0,DLY1 f	
00455	0f15	080413		FCB	G5,DS5,DLY2 m	
00456	0f18	060009		FCB	F5,0,DLY1 r	
00457	0f1b	040009		FCB	DS5,0,DLY1 d	
00458	0f1e	040b13	FINE1	FCB	DS5,AS5,DLY2 d	
00459	0f21	040809		FCB	DS5,G5,DLY1 d	
00460	0f24	060009		FCB	F5,0,DLY1 r	
00461	0f27	080b13		FCB	G5,AS5,DLY2 m	
00462	0f2a	060009		FCB	F5,0,DLY1 r	
00463	0f2d	001909		FCB	0,GS6,DLY1 f	
00464	0f30	060913	FINE2	FCB	F5,GS5,DLY2 r	
00465	0f33	080b09		FCB	G5,AS5,DLY1 m	
00466	0f36	060009		FCB	F5,0,DLY1 r	
00467	0f39	040b09		FCB	DS5,AS5,DLY1 d	
00468	0f3c	060009		FCB	F5,0,DLY1 r	
00469	0f3f	04003b		FCB	DS5,0,DLY6 d	
00470		*				
00471	1ffe			ORG	\$1FFE	
00472	1ffe	0d00	RSTV	FDB	START	
00473	1ff8			ORG	\$1FF8	
00474	1ff8	0d46	KEYV	FDB	KEYSCN	
00475				END		

Motorola reserves the right to make changes without further notice to any products herein to improve reliability, function or design. Motorola does not assume any liability arising out of the application or use of any product or circuit described herein; neither does it convey any license under its patent rights nor the rights of others. Motorola and  are registered trademarks of Motorola, Inc. Motorola, Inc. is an Equal Employment Opportunity/Affirmative Action Employer.

Asia Pacific Literature Distribution Centers :

Motorola Semiconductors Hong Kong Ltd., P.O. Box 80300; Cheung Sha Wan Post Office; Kowloon Hong Kong.

